МІНІСТЕРСТВО ОСВІТИ І НАУКИ УКРАЇНИ
НАЦІОНАЛЬНИЙ УНІВЕРСИТЕТ «ЛЬВІВСЬКА ПОЛІТЕХНІКА»

МЕТОД ПРЯМИХ ЧИСЕЛЬНОГО РОЗВ’ЯЗУВАННЯ РІВНЯНЬ ПАРАБОЛІЧНОГО ТИПУ

Методичні вказівки

до лабораторної роботи з курсу «Чисельні методи математичної фізики»
для студентів базового напрямку
6.0802 «Прикладна математика»

Затверджено
на засіданні кафедри
прикладної математики.
Протокол № 8 від 23.01.2014

Львів – 2014

Метод прямих чисельного розв’язування рівнянь параболічного типу: Методичні вказівки до лабораторної роботи з курсу «Чисельні методи математичної фізики» для студентів базового напрямку 6.0802 «Прикладна математика» / Укл.: М. В. Кутнів, О. І. Паздрій — Львів: Видавництво Національного університету «Львівська політехніка», 2014. — 21 с.

Укладачі Кутнів М. В., доктор фіз.-мат. наук, доц.
Паздрій О. І., асист.

Відповідальний за випуск Костробій П. П., доктор фіз.-мат. наук, проф.

Рецензенти Демків І. І., кандидат фіз.-мат. наук, доц.
Пізюр Я. В., кандидат фіз.-мат. наук, доц.

ТЕОРЕТИЧНА ЧАСТИНА

Метод прямих чисельного розв'язування рівнянь параболічного типу

Розглянемо параболічне диференціальне рівняння з частинними похідними

$$\frac{\partial u}{\partial t}=\frac{\partial}{\partial x}\left(k(x,t)\frac{\partial u}{\partial x}\right)+f(x,t),\quad 0<x<l,\quad 0<t\le T, \tag{1}$$

розв'язок якого задовольняє початкову умову

$$u(x,0)=\varphi(x),\quad 0\le x\le l \tag{2}$$

та крайові умови

$$u(0,t)=\mu_1(t),\quad u(l,t)=\mu_2(t),\qquad 0\le t\le T. \tag{3}$$

Коефіцієнт $k(x,t)$ обмежений знизу і зверху, тобто

$$0<c_1\le k(x,t)\le c_2,\quad 0\le x\le l,\quad 0\le t\le T,$$

де c_1, c_2 — сталі.

Припустимо, що задача (1) – (3) має єдиний розв'язок, який має потрібну кількість похідних.

Крайову задачу (1) – (3) будемо розв'язувати чисельно за допомогою методу прямих [2,8]. На інтервалі $0\le x\le l$ введемо рівномірну сітку $\overline{\omega}_\Delta=\left\{x_i=i\Delta,\, i=\overline{0,N},\, \Delta=l/N\right\}$. Для апроксимації диференціального оператора по просторовій змінній використаємо інтегро-інтерполяційний метод [6]. Проінтегруємо рівняння (1) на відрізку $x_{i-1/2}\le x\le x_{i+1/2}$, $x_{i\pm1/2}=x_i\pm0{,}5\Delta$, тоді отримаємо

$$\int\limits_{x_{i-1/2}}^{x_{i+1/2}}\frac{\partial u}{\partial t}dx=w(x_{i+1/2},t)-w(x_{i-1/2},t)+\int\limits_{x_{i-1/2}}^{x_{i+1/2}}f(x,t)dx, \tag{4}$$

де

$$w(x,t)=k(x,t)\frac{\partial u}{\partial x}.$$

Апроксимуємо інтеграли та похідні, які входять в рівняння балансу (4)

$$\int\limits_{x_{i-1/2}}^{x_{i+1/2}}\frac{\partial u}{\partial t}dx\approx\Delta\frac{\partial u(x_i,t)}{\partial t},\qquad \int\limits_{x_{i-1/2}}^{x_{i+1/2}}f(x,t)dx\approx\Delta f(x_i,t).$$

Тоді замість (4) отримаємо рівняння

$$\frac{\partial u(x_i,t)}{\partial t}=\frac{w(x_{i+1/2},t)-w(x_{i-1/2},t)}{\Delta}+f(x_i,t). \tag{5}$$

Введемо позначення $u_i(t) = u(x_i, t)$. Виразимо $w(x_{i\pm 1/2}, t)$ через значення $u(x,t)$ в вузлах сітки $\overline{\omega}_\Delta$. Для цього проінтегруємо рівняння

$$\frac{\partial u}{\partial x} = \frac{w(x,t)}{k(x,t)}$$

на відрізку $x_{i-1} \le x \le x_i$:

$$u_i(t) - u_{i-1}(t) = \int_{x_{i-1}}^{x_i} \frac{w(x,t)}{k(x,t)} dx \approx w(x_{i-1/2}, t) \int_{x_{i-1}}^{x_i} \frac{dx}{k(x,t)}. \tag{6}$$

Нехай

$$\left(\frac{1}{\Delta} \int_{x_{i-1}}^{x_i} \frac{dx}{k(x,t)} \right)^{-1} \approx k(x_{i-1/2}, t) = a_i(t), \tag{7}$$

тоді з (6) випливає рівність

$$w(x_{i-1/2}, t) \approx a_i(t) \frac{u_i(t) - u_{i-1}(t)}{\Delta}.$$

Підставляючи вирази для $w(x_{i\pm 1/2}, t)$ в (5), отримаємо систему звичайних диференціальних рівнянь (ЗДР)

$$\begin{gathered} \frac{dy_i}{dt} = \frac{1}{\Delta} \left(a_{i+1}(t) \frac{y_{i+1}(t) - y_i(t)}{\Delta} - a_i(t) \frac{y_i(t) - y_{i-1}(t)}{\Delta} \right) + \\ + f(x_i, t), \quad i = \overline{1, N-1}, \quad t \in (0, T], \end{gathered} \tag{8}$$

де $y_i(t) \approx u_i(t) = u(x_i, t)$, $a_i(t)$ визначаються рівністю (7) або

$$a_i(t) = \frac{1}{2} \big(k(x_i, t) + k(x_{i-1}, t) \big). \tag{9}$$

Приєднаємо до системи рівнянь (8) крайові умови:

$$y_0(t) = \mu_1(t), \quad y_N(t) = \mu_2(t),$$

за допомогою яких виключимо значення $y_0(t), y_N(t)$ з рівностей (8) для $i = 1, i = N-1$, тоді отримаємо задачу Коші для системи ЗДР

$$\begin{gathered} \frac{dy_1}{dt} = \frac{1}{\Delta} \left(a_2(t) \frac{y_2(t) - y_1(t)}{\Delta} - a_1(t) \frac{y_1(t) - \mu_1(t)}{\Delta} \right) + f(x_1, t), \\ \frac{dy_i}{dt} = \frac{1}{\Delta} \left(a_{i+1}(t) \frac{y_{i+1}(t) - y_i(t)}{\Delta} - a_i(t) \frac{y_i(t) - y_{i-1}(t)}{\Delta} \right) + \\ + f(x_i, t), \quad i = \overline{2, N-2}, \quad t \in (0, T], \end{gathered} \tag{10}$$

$$\frac{dy_{N-1}}{dt}=\frac{1}{\Delta}\left(a_N(t)\frac{\mu_2(t)-y_{N-1}(t)}{\Delta}-a_{N-1}(t)\frac{y_{N-1}(t)-y_{N-2}(t)}{\Delta}\right)+$$
$$+f(x_{N-1},t),$$
$$y_i(0)=\varphi(x_i),\quad i=\overline{1,N-1}.$$

Нехай розв’язок диференціального рівняння (1) задовольняє крайові умови

$$k(0,t)\frac{\partial u(0,t)}{\partial x}=\beta_1(t)u(0,t)-\mu_1(t),\quad \beta_1(t)>0, \tag{11}$$

$$-k(l,t)\frac{\partial u(l,t)}{\partial x}=\beta_2(t)u(l,t)-\mu_2(t),\quad \beta_2(t)>0. \tag{12}$$

Побудуємо різницевий аналог цих умов. Для цього, згідно з інтегро-інтерполяційним методом, проінтегруємо рівняння (1) на відрізку $0\le x\le x_{1/2}$, де $x_{1/2}=0,5\Delta$:

$$\int_0^{x_{1/2}}\frac{\partial u}{\partial t}dx=w(x_{1/2},t)-w(0,t)+\int_0^{x_{1/2}}f(x,t)dx.$$

Враховуючи рівності

$$\int_0^{x_{1/2}}\frac{\partial u}{\partial t}dx\approx\frac{\Delta}{2}\frac{\partial u(0,t)}{dt},\quad \int_0^{x_{1/2}}f(x,t)dx\approx\frac{\Delta}{2}f(x_0,t),$$

$$w(x_{1/2},t)\approx a_1(t)\frac{u_1(t)-u_0(t)}{\Delta},\quad w(0,t)=k(0,t)\frac{\partial u(0,t)}{\partial x}=\beta_1(t)u(0,t)-\mu_1(t),$$

отримаємо різницеву крайову умову

$$\frac{dy_0}{dt}=\frac{2}{\Delta}\left(a_1(t)\frac{y_1(t)-y_0(t)}{\Delta}-\beta_1(t)y_0(t)+\mu_1(t)\right)+f(x_0,t).$$

Різницевий аналог крайової умови (12), який також можна побудувати інтегро-інтерполяційним методом, буде мати вигляд

$$\frac{dy_N}{dt}=\frac{2}{\Delta}\left(\mu_2(t)-\beta_2(t)y_N(t)-a_N(t)\frac{y_N(t)-y_{N-1}(t)}{\Delta}\right)+f(x_N,t).$$

Похибка апроксимації диференціальних операторів, які входять в диференціальне рівняння та крайові умови, відповідними різницевими операторами є величиною порядку $O(\Delta^2)$.

Отже, наближений розв’язок $y_i(t)$, $i=\overline{0,N}$ задачі (1), (2), (11), (12) є розв’язком задачі Коші

$$
\begin{aligned}
&\frac{dy_0}{dt}=\frac{2}{\Delta}\left(a_1(t)\frac{y_1(t)-y_0(t)}{\Delta}-\beta_1(t)y_0(t)+\mu_1(t)\right)+f(x_0,t),\\
&\frac{dy_i}{dt}=\frac{1}{\Delta}\left(a_{i+1}(t)\frac{y_{i+1}(t)-y_i(t)}{\Delta}-a_i(t)\frac{y_i(t)-y_{i-1}(t)}{\Delta}\right)+\\
&\qquad\qquad+f(x_i,t),\quad i=\overline{1,N-1},\quad t\in(0,T],\\
&\frac{dy_N}{dt}=\frac{2}{\Delta}\left(\mu_2(t)-\beta_2(t)y_N(t)-a_N(t)\frac{y_N(t)-y_{N-1}(t)}{\Delta}\right)+f(x_N,t),\\
&\qquad\qquad y_i(0)=\varphi(x_i),\quad i=\overline{0,N},
\end{aligned}
\tag{13}
$$

де коефіцієнти $a_i(t)$ обчислюють за формулою (7) або (9).

Розглянемо тепер крайову задачу для квазілінійного рівняння теплопровідності. Нехай розв'язок рівняння

$$
\frac{\partial u}{\partial t}=\frac{\partial}{\partial x}\left(k(x,t,u)\frac{\partial u}{\partial x}\right)+f\left(x,t,u,\frac{\partial u}{\partial x}\right),\quad 0<x<l,\quad 0\le t\le T,
$$

задовольняє початкову умову (2) та крайові умови

$$
\alpha_1(t)\frac{\partial u(0,t)}{\partial x}=\beta_1(t)u(0,t)-\mu_1(t),\quad 0<t\le T,
$$

$$
-\alpha_2(t)\frac{\partial u(l,t)}{\partial x}=\beta_2(t)u(l,t)-\mu_2(t),\quad 0<t\le T,
$$

де $\alpha_1(t)>0$, $\alpha_2(t)>0$.

Методом прямих зведемо задачу до задачі Коші для системи нелінійних ЗДР вигляду

$$
\begin{aligned}
&\frac{dy_0}{dt}=\frac{2}{\Delta}\left(a_1(t,y)\frac{y_1(t)-y_0(t)}{\Delta}-k_0\frac{\beta_1(t)y_0(t)-\mu_1(t)}{\alpha_1(t)}\right)+\\
&\qquad\qquad+f\left(x_0,t,y_0,\frac{\beta_1(t)y_0(t)-\mu_1(t)}{\alpha_1(t)}\right),\\
&\frac{dy_i}{dt}=\frac{1}{\Delta^2}\Big(a_{i+1}(t,y)\big(y_{i+1}(t)-y_i(t)\big)-a_i(t)\big(y_i(t)-y_{i-1}(t)\big)\Big)+\\
&\qquad\qquad+f\left(x_i,t,y_i,\frac{y_{i+1}-y_{i-1}}{2\Delta}\right),\quad i=\overline{1,N-1},\\
&\frac{dy_N}{dt}=\frac{2}{\Delta}\left(k_N\frac{\mu_2(t)-\beta_2(t)y_N(t)}{\alpha_2(t)}-a_N(t,y)\frac{y_N(t)-y_{N-1}(t)}{\Delta}\right)+\\
&\qquad\qquad+f\left(x_N,t,y_N,\frac{\mu_2(t)-\beta_2(t)y_N(t)}{\alpha_2(t)}\right),
\end{aligned}
\tag{14}
$$

$$y_i(0) = \varphi(x_i) \quad i = \overline{0, N}.$$

де $\quad k_0 = k(0,t,y_0), \quad k_N = k(l,t,y_N), \qquad a_i(t,y) = k\left(x_{i-1/2}, t, \dfrac{y_i + y_{i-1}}{2}\right), \qquad$ або

$a_i(t) = \dfrac{1}{2}\big(k(x_{i-1}, t, y_{i-1}) + k(x_i, t, y_i)\big).$

Зауважимо, що матриця Якобі правих частин системи ЗДР (10) або (13) або (14) має стрічкову, тридіагональну структуру. Крім того, задача Коші є жорсткою задачею (див., напр., [2], стр. 25), причому при зростанні N коефіцієнт жорсткості задачі зростає. Отже, для розв'язування задачі Коші для системи ЗДР будемо використовувати формули диференціювання назад (див., напр., [4,9]).

Приклад 1. Нехай необхідно розв'язати крайову задачу

$$\frac{\partial u}{\partial t} = \frac{\partial^2 u}{\partial x^2} - 128xt, \quad 0 < x < 1, \quad t \in (0;1], \tag{15}$$

$$u(x,0) = 4x^2, \quad 0 \le x \le 1, \tag{16}$$

$$\frac{\partial u(0,t)}{\partial x} = 16t, \quad \frac{\partial u(1,t)}{\partial x} + 2u(1,t) = 0. \tag{17}$$

Для зведення крайової задачі (15) – (17) до задачі Коші для системи ЗДР використаємо метод заміни похідних різницевими співвідношеннями (див., напр., [1,4,6,7]). У диференціальному рівнянні (15) частинні похідні за змінною x у вузлі сітки x_i замінимо другою різницевою похідною

$$\frac{\partial^2 u(x_i,t)}{\partial x^2} = \frac{u_{i+1} - 2u_i + u_{i-1}}{\Delta^2} + O(\Delta^2),$$

яка має другий порядок апроксимації, тоді отримаємо

$$\frac{dy_i}{dt} = \frac{y_{i+1}(t) - 2y_i(t) + y_{i-1}(t)}{\Delta^2} - 128x_i t, \quad i = \overline{1, M-1},$$

де $y_i(t) \approx u(x_i,t) = u_i(t)$. Для апроксимації крайових умов (17) з другим порядком застосуємо метод зменшення похибки апроксимації (див., наприклад, [6], стр. 84). Оскільки

$$u_{x,0} = \frac{\partial u(0,t)}{\partial x} + \frac{\Delta}{2}\frac{\partial^2 u(0,t)}{\partial x^2} + O(\Delta^2),$$

то, з урахуванням диференціального рівняння (15) отримаємо

$$\frac{\partial u(0,t)}{\partial x} = u_{x,0} - \frac{\Delta}{2}\frac{\partial u(0,t)}{\partial t} + O(\Delta^2) = 16t,$$

Звідси випливає рівняння

$$\frac{dy_0}{dt}=\frac{2\left(y_1(t)-y_0(t)\right)}{\Delta^2}-\frac{32t}{\Delta}.$$

Оскільки

$$u_{\bar{x},M}=\frac{\partial u(1,t)}{\partial x}-\frac{\Delta}{2}\frac{\partial^2 u(1,t)}{\partial x^2}+O(\Delta^2),$$

то враховуючи (15) будемо мати

$$u_{\bar{x},M}=\frac{\partial u(1,t)}{\partial x}-\frac{\Delta}{2}\left(\frac{\partial u(1,t)}{\partial t}+128t\right)+O(\Delta^2),$$

Підставимо це значення в крайову умову (17), тоді отримаємо

$$u_{\bar{x},M}+\frac{\Delta}{2}\left(\frac{\partial u(1,t)}{\partial t}+128t\right)+O(\Delta^2)+2u(1,t)=0.$$

Звідси

$$\frac{dy_M}{dt}=-\frac{2\left(y_M(t)-y_{M-1}(t)\right)}{\Delta^2}-\frac{4y_M(t)}{\Delta}-128t.$$

Перенумеруємо рівняння системи за допомогою заміни $k=i+1$, тоді $k=\overline{1,N}$, $N=M+1$, тоді отримаємо задачу Коші для системи ЗДР

$$\begin{aligned}
&\frac{dy_1}{dt}=\frac{2\left(y_2(t)-y_1(t)\right)}{\Delta^2}-\frac{32t}{\Delta},\\
&\frac{dy_k}{dt}=\frac{y_{k+1}(t)-2y_k(t)+y_{k-1}(t)}{\Delta^2}-128x_{k-1}t,\quad k=\overline{2,N-1},\\
&\frac{dy_N}{dt}=-\frac{2\left(y_N(t)-y_{N-1}(t)\right)}{\Delta^2}-\frac{4y_N(t)}{\Delta}-128t,
\end{aligned}\tag{18}$$

$$y_k(0)=4x_{k-1}^2,\quad k=\overline{1,N},\tag{19}$$

де $x_{k-1}=(k-1)\Delta,\ \Delta=1/(N-1)$.

Зауважимо, що задача Коші (18), (19) апроксимує задачу (15) – (17) з другим порядком.

Задачу Коші будемо розв’язувати чисельно за допомогою програми STIFF [5]. Головну програму та підпрограму DIFFUN наведено в Додатку 1. Результати розв’язування задачі з точністю $EPS=10^{-2}$ в режимі $MF=25$ в точці $t=1$:

$y_1=-0,3098\cdot10^2,\quad y_2=-0,3022\cdot10^2,\quad y_3=-0,2958\cdot10^2,\quad y_4=-0,2904\cdot10^2,$
$y_5=-0,2859\cdot10^2,\quad y_6=-0,2820\cdot10^2,\quad y_7=-0,2787\cdot10^2,\quad y_8=-0,2755\cdot10^2,$
$y_9=-0,2724\cdot10^2,\quad y_{10}=-0,2691\cdot10^2,\quad y_{11}=-0,2653\cdot10^2,\quad y_{12}=-0,2611\cdot10^2,$
$y_{13}=-0,2560\cdot10^2,\quad y_{14}=-0,2498\cdot10^2,\quad y_{15}=-0,2424\cdot10^2,\quad y_{16}=-0,2335\cdot10^2,$
$y_{17}=-0,2228\cdot10^2,\quad y_{18}=-0,2102\cdot10^2,\quad y_{19}=-0,1952\cdot10^2,\quad y_{20}=-0,1779\cdot10^2.$

$NSTEP=43,\qquad NFUN=301,\qquad NJAC=11.$

Приклад 2. Розв’язати методом прямих задачу

$$\frac{\partial u}{\partial t}=\frac{\partial^2 u}{\partial x^2}-c\frac{\partial u}{\partial x},\quad 0<x<1,\quad 0<t\le 0,04, \tag{20}$$

$$u(x,0)=0,\quad 0\le x\le 1, \tag{21}$$

$$u(0,t)=1,\quad \frac{\partial u(1,t)}{\partial x}=0,\quad 0\le t\le 0,04. \tag{22}$$

Замінимо в рівнянні (20) частинні похідні за змінною x у вузлі x_i відповідними різницевими похідними

$$\frac{\partial^2 u(x_i,t)}{\partial x^2}=\frac{u_{i+1}-2u_i+u_{i-1}}{\Delta^2}+O(\Delta^2),\quad \frac{\partial u(x_i,t)}{\partial x}=\frac{u_{i+1}-u_{i-1}}{2\Delta}+O(\Delta^2),$$

тоді отримаємо

$$\frac{dy_i}{dt}=\frac{y_{i+1}(t)-2y_i(t)+y_{i-1}(t)}{\Delta^2}-c\frac{y_{i+1}(t)-y_{i-1}(t)}{2\Delta},\quad i=\overline{1,M}. \tag{23}$$

Для апроксимації крайової умови в точці $x=1$ розглянемо метод фіктивних точок ([6], стр. 306). Введемо зовні відрізка $0\le x\le 1$ фіктивну точку $x_{M+1}=x_M+\Delta$ і будемо вважати, що вихідне рівняння задовольняється при $x\le x_{M+1}$. Різницеві аналоги крайових умов (22) будуть мати вигляд

$$y_0(t)=1,\quad \frac{y_{M+1}-y_{M-1}}{2\Delta}=0\,. \tag{24}$$

Використовуючи (24), виключимо з (23) для $i=1,\ i=M$ значення $y_0,\ y_{M+1}$, тоді отримаємо задачу Коші для системи ЗДР

$$\frac{dy_1}{dt}=\frac{y_2(t)-2y_1(t)+1}{\Delta^2}-c\frac{y_2(t)-1}{2\Delta},$$

$$\frac{dy_i}{dt}=\frac{y_{i+1}(t)-2y_i(t)+y_{i-1}(t)}{\Delta^2}-c\frac{y_{i+1}(t)-y_{i-1}(t)}{2\Delta},\quad i=\overline{2,M-1},$$

$$\frac{dy_M}{dt}=-2\frac{y_M(t)-y_{M-1}(t)}{\Delta^2},$$

$$y_i(0)=0,\quad i=\overline{1,M}.$$

Цю задачу Коші розв'яжемо за допомогою програми STIFF (див. Додаток 2) з точністю $\varepsilon = 10^{-2}$ в режимі $MF = 25$. Результати розв'язування в точці $t = 0,04$:

$$y_1 = 0,9999, \quad y_2 = 0,9999, \quad y_3 = 0,9998, \quad y_4 = 0,9995, \quad y_5 = 0,9990,$$
$$y_6 = 0,9980, \quad y_7 = 0,9960, \quad y_8 = 0,9927, \quad y_9 = 0,9871, \quad y_{10} = 0,9785,$$
$$y_{11} = 0,9656, \quad y_{12} = 0,9470, \quad y_{13} = 0,9226, \quad y_{14} = 0,8904, \quad y_{15} = 0,8506,$$
$$y_{16} = 0,8032, \quad y_{17} = 0,7487, \quad y_{18} = 0,6888, \quad y_{19} = 0,6285, \quad y_{20} = 0,5852.$$
$$NSTEP = 36, \qquad NFUN = 189, \qquad NJAC = 6.$$

Послідовність виконання лабораторної роботи

1. Одержати варіант завдання (див. Додаток 3).
2. Вивчити відповідний лекційний матеріал та методичні вказівки до цієї лабораторної роботи.
3. Звести крайову задачу для рівняння параболічного типу до задачі Коші для системи ЗДР.
4. Для чисельного розв’язування задачі Коші для системи ЗДР за допомогою програми *STIFF* підготувати головну програму та підпрограму-процедуру *DIFFUN*. Провести відлагодження програми і отримати розв’язок задачі при $MF = 25$ з точністю $EPS = 10^{-2}$.
5. Вивести на друк в точці *TEND* значення чисельного розв’язку, а також основні характеристики програми *NSTEP*, *NFUN*, *NJAC*.

Зміст звіту

1. Постановка крайової задачі для рівняння параболічного типу.
2. Задача Коші для системи ЗДР, отримана в результаті зведення крайової задачі для рівняння параболічного типу.
3. Текст головної програми та підпрограми *DIFFUN*.
4. Результати розв′язування задачі.

Література

1. Гаврилюк І. П., Макаров В. Л. Методи обчислень. — К.: Вища школа, 1995, ч. 1, ч.2.
2. Деккер К., Вервер Я. Устойчивость методов Рунге-Кутты для жестких нелинейных дифференциальных уравнений, — М.: Мир, 1990.
3. Калиткин Н. Н. Численные методы. — М.:Наука, 1978.
4. Кутнів М. В. Чисельні методи: Навчальний посібник. — Львів, «Растр–7», 2010.
5. Лінійні багатокрокові методи чисельного розв'язування задачі Коші для звичайних диференціальних рівнянь: Методичні вказівки до лабораторної роботи з курсу «Чисельні методи» для студентів базового напрямку 6.0802 «Прикладна математика», / Укл.: М. В. Кутнів, Я. В. Пізюр, Є. М. Максимів — Львів: Видавництво Національного університету «Львівська політехніка», 2008. — 40 с.
6. Самарский А. А. Теория разностных схем. — М.: Наука, 1989.
7. Самарский А. А., Гулин А. В. Численные методы. — М.: Наука, 1986.
8. Современные численные методы решения обыкновенных дифференциальных уравнений // Под ред. Дж. Холла, Дж. Уатта. — М.: Мир, 1979. — 312 с.
9. Хайрер Э., Нерсетт С., Ваннер Г. Решение обыкновенных дифференциальных уравнений. Нежесткие задачи. — М.: Мир, 1990.

Додаток 1. Приклад 1

```
C      МЕТОД ПРЯМИХ ЧИСЕЛЬНОГО РОЗВ'ЯЗУВАННЯ РІВНЯНЬ
C      ПАРАБОЛІЧНОГО ТИПУ
       IMPLICIT REAL*8(A-H,O-Z)
       DIMENSION Y(20,13),YMAX(20),ERROR(20),PW(80),
      1          FSAVE(40),IWORK(20)
       COMMON/STCOM1/T,H,HMIN,HMAX,EPS,N,MF,KFLAG,JSTART,MAXORD
       COMMON/STCOM2/HUSED,NQUSED
       COMMON/STCOM3/ML,MU
       COMMON/STCOM4/NSTEP,NFUN,NJAC
       COMMON/PCOM/COEF,DX
       NYDIM=20
       EPS=1.D-2
       KB=3
 401   CONTINUE
       N=20
       T=0.0D0
       TEND=1.0D0
       H=1.D-3
       DX=1.D0/DBLE(N-1)
       COEF=1.D0/DX**2
       DO I=1,N
         XIM1=(I-1)*DX
         Y(I,1)=4.D0*XIM1**2
       END DO
       HMAX=TEND
       HMIN=1.D-15
       JSTART=0
       MF=25
       ML=1
       MU=1
       MAXORD=5
       WRITE(0,20) MF,EPS
 20    FORMAT(//3X,'MF=',I2/,' EPS='D11.3)
       NSTEP=0
       NFUN=0
       NJAC=0
       DO I=1,N
         YMAX(I)=DMAX1(DABS(Y(I,1)),1.D0)
       END DO
 40    CONTINUE
       DO I=1,N
         YMAX(I)=DMAX1(DABS(Y(I,1)),YMAX(I))
       END DO
       CALL STIFF(Y,YMAX,ERROR,PW,FSAVE,IWORK,NYDIM)
       IF(KFLAG.EQ.0)GO TO 60
       WRITE(0,50) KFLAG
 50    FORMAT(/'  KFLAG=',I2/)
       STOP
```

```
60    CONTINUE
      IF(DABS(TEND-T).LE.1.D-15) GO TO 90
      IF(TEND-T-H.GE.0.D0) GO TO 40
      E=TEND-T
      S=E/H
      DO I=1,N
        DO J=1,JSTART
          Y(I,1)=Y(I,1)+Y(I,J+1)*S**J
        END DO
      END DO
      T=T+E
      GO TO 60
90    CONTINUE
      WRITE(0,556) H,T,(Y(I,1),I=1,N)
556   FORMAT(1X,5D16.8)
      WRITE(0,95) NSTEP,NFUN,NJAC
95    FORMAT(/'  NSTEP=',I4,' NFUN= ',I5,' NJAC=',I4)
      KB=KB+1
      IF(KB.GE.3) GO TO 402
      EPS=EPS*1.D-2
      GO TO 401
402   CONTINUE
      STOP
      END
      SUBROUTINE DIFFUN (N,T,Y,YDOT)
      IMPLICIT REAL*8 (A-H,O-Z)
      DIMENSION Y(1),YDOT(1)
      COMMON/PCOM/ COEF,DX
      YDOT(1)=2.D0*COEF*(Y(2)-Y(1))-3.2D1*T/DX
      YDOT(N)=-2.D0*COEF*(Y(N)-Y(N-1))-4.0D0*Y(N)/DX-1.28D2*T
      DO I=2,N-1
        XIM1=(I-1)*DX
        YDOT(I)=COEF*(Y(I+1)-2.D0*Y(I)+Y(I-1))-1.28D2*XIM1*T
      END DO
      RETURN
      END
```

Додаток 2. Приклад 2

```
C      МЕТОД ПРЯМИХ ЧИСЕЛЬНОГО РОЗВ'ЯЗУВАННЯ РІВНЯНЬ
C      ПАРАБОЛІЧНОГО ТИПУ
       IMPLICIT REAL*8(A-H,O-Z)
       DIMENSION Y(20,13),YMAX(20),ERROR(20),PW(80),
      1          FSAVE(40),IWORK(20)
       COMMON/STCOM1/T,H,HMIN,HMAX,EPS,N,MF,KFLAG,JSTART,MAXORD
       COMMON/STCOM2/HUSED,NQUSED
       COMMON/STCOM3/ML,MU
       COMMON/STCOM4/NSTEP,NFUN,NJAC
       COMMON/PCOM/COEF1,COEF2,NM1
       C=0.25D2
       NYDIM=20
       EPS=1.D-2
       KB=3
 401   CONTINUE
       N=20
       T=0.0D0
       TEND=0.4D-1
       H=1.D-3
       DO 10 I=1,N
 10    Y(I,1)=0.D0
       HMAX=TEND
       HMIN=1.D-15
       DX=1.D0/DBLE(N)
       COEF1=1.D0/DX**2
       COEF2=0.5D0*C/DX
       NM1=N-1
       JSTART=0
       MF=25
       ML=1
       MU=1
       MAXORD=5
       WRITE(0,20) MF,EPS
 20    FORMAT(//3X,'MF=',I2/,' EPS='D11.3)
       NSTEP=0
       NFUN=0
       NJAC=0
       DO 30 I=1,N
 30    YMAX(I)=DMAX1(DABS(Y(I,1)),1.D0)
 40    CONTINUE
       DO 45 I=1,N
 45    YMAX(I)=DMAX1(DABS(Y(I,1)),YMAX(I))
       CALL STIFF(Y,YMAX,ERROR,PW,FSAVE,IWORK,NYDIM)
       IF(KFLAG.EQ.0)GO TO 60
       WRITE(0,50) KFLAG
 50    FORMAT(/'  KFLAG=',I2/)
       STOP
 60    CONTINUE
```

```
      IF(DABS(TEND-T).LE.1.D-15) GO TO 90
      IF(TEND-T-H) 80,40,40
80    E=TEND-T
      S=E/H
      DO 85 I=1,N
      DO 85 J=1,JSTART
85    Y(I,1)=Y(I,1)+Y(I,J+1)*S**J
      T=T+E
      GO TO 60
90    CONTINUE
      WRITE(0,556) H,T,(Y(I,1),I=1,N)
556   FORMAT(1X,5D16.8)
      WRITE(0,95) NSTEP,NFUN,NJAC
95    FORMAT(/'  NSTEP=',I4,' NFUN= ',I5,' NJAC=',I4)
      KB=KB+1
      IF(KB.GE.3) GO TO 402
      EPS=EPS*1.D-2
      GO TO 401
402   CONTINUE
      STOP
      END
      SUBROUTINE DIFFUN (N,T,Y,YDOT)
      IMPLICIT REAL*8 (A-H,O-Z)
      DIMENSION Y(1),YDOT(1)
      COMMON/PCOM/ COEF1,COEF2,NM1
      YDOT(1)=COEF1*(1.D0-2.D0*Y(1)+Y(2))-
     1        COEF2*(Y(2)-1.D0)
      YDOT(N)=2.D0*COEF1*(Y(NM1)-Y(N))
      DO 10 I=2,NM1
      YDOT(I)=COEF1*(Y(I+1)-2.D0*Y(I)+Y(I-1))
     1         -COEF2*(Y(I+1)-Y(I-1))
10    CONTINUE
      RETURN
      END
```

Додаток 3. Варіанти завдань

1. $$\frac{\partial u}{\partial t}=\frac{\partial^2 u}{\partial x^2}+u+xt(2-t)+2\cos t,\quad 0<x<\pi,\quad t\in(0,1],$$
$$u(x,0)=\cos 2x,\quad 0\le x\le \pi,$$
$$\frac{\partial u(0,t)}{\partial x}=t^2,\quad \frac{\partial u(\pi,t)}{\partial x}=t^2,\quad t\in[0,1].$$

2. $$\frac{\partial u}{\partial t}=\frac{\partial^2 u}{\partial x^2}+9u+4\sin^2 t\cos 3x-9x^2-2,\quad 0<x<\pi,\quad t\in(0,1],$$
$$u(x,0)=x^2+2,\quad 0\le x\le \pi,$$
$$\frac{\partial u(0,t)}{\partial x}=0,\quad \frac{\partial u(\pi,t)}{\partial x}=2\pi,\quad t\in[0,1].$$

3. $$\frac{\partial u}{\partial t}=\frac{\partial^2 u}{\partial x^2}+3t\sin x,\quad 0<x<1,\quad t\in(0,1],$$
$$u(x,0)=e^{-0,1x}\sin\frac{\pi}{12}x,\quad 0\le x\le 1,$$
$$u(0,t)=0,\quad u(1,t)=e^{-0,1}\sin\frac{\pi}{12},\quad t\in[0,1].$$

4. $$\frac{\partial u}{\partial t}=a\frac{\partial^2 u}{\partial x^2}-u\frac{\partial u}{\partial x},\quad 0<x<1,\quad t\in(0,4],$$
$$u(x,0)=\left(1+e^{\frac{x}{2a}}\right)^{-1},\quad 0\le x\le 1,$$
$$u(0,t)=\left(1+e^{-\frac{t}{4a}}\right)^{-1},\quad u(1,t)=\left(1+e^{\frac{2a-t}{4a}}\right)^{-1},\quad t\in\left[0,4\right],$$
$$u(x,t)=\left(1+e^{\frac{x}{2a}-\frac{t}{4a}}\right)^{-1},\quad a=0,05.$$

5. $$\frac{\partial u}{\partial t}=\frac{\partial}{\partial x}\left((0,1+\cos^2 x)\frac{\partial u}{\partial x}\right),\quad 0<x<1,\quad t\in(0,1],$$
$$u(x,0)=1-x,\quad 0\le x\le 1,$$
$$u(0,t)=\cos t,\quad u(1,t)=0,\quad t\in\left[0,1\right].$$

6. $$\frac{\partial u}{\partial t}=\frac{1}{2}\frac{\partial^2 u}{\partial x^2}-xt,\quad 0<x<2,\quad t\in(0,1],$$
$$u(x,0)=x^2,\quad 0\le x\le 2,$$
$$\frac{\partial u(0,t)}{\partial x}=t,\quad \frac{\partial u(2,t)}{\partial x}+u(2,t)=0,\quad t\in\left[0,1\right].$$

7. $$\frac{\partial u}{\partial t}=\frac{\partial^2 u}{\partial x^2}+8x^2t^4, \quad 0<x<1, \quad t\in(0,1],$$
$$u(x,0)=\sin(\pi x), \quad 0\le x\le 1,$$
$$\frac{\partial u(0,t)}{\partial x}-0,5u(0,t)=0, \quad \frac{\partial u(1,t)}{\partial x}+0,5u(1,t)=0, \quad t\in[0,1].$$

8. $$\frac{\partial u}{\partial t}=a\frac{\partial^2 u}{\partial x^2}-0,9au^2e^{x-t}-au, \quad 0<x<\frac{1}{2}, \quad t\in(0,\,0,5],$$
$$u(x,0)=10e^{-x}, \quad 0\le x\le\frac{1}{2},$$
$$u(0,t)=(1-0,9e^{-at})^{-1}, \quad u\left(\frac{1}{2},t\right)=e^{-0.5}(1-0,9e^{-at})^{-1}, \quad t\in[0,\,0,5],$$
$$a=1, \quad u(x,t)=e^{-x}(1-0,9e^{-at})^{-1}.$$

9. $$\frac{\partial u}{\partial t}=c\frac{\partial^2 u}{\partial x^2}-2cu, \quad 0<x<\frac{1}{2}, \quad t\in(0,1],$$
$$u(x,0)=e^{x}, \quad 0\le x\le\frac{1}{2},$$
$$u(0,t)=e^{-t}, \quad u\left(\frac{1}{2},t\right)=e^{\frac{1}{2}-t}, \quad t\in[0,1],$$
$$c=1, \quad u(x,t)=e^{x-ct}.$$

10. $$\frac{\partial u}{\partial t}=c\frac{\partial^2 u}{\partial x^2}+4\frac{\partial u}{\partial x}+x-4t+1+e^{-2x}\cos^2(\pi x), \quad 0<x<1, \quad t\in(0,1],$$
$$u(x,0)=0, \quad 0\le x\le 1,$$
$$u(0,t)=t, \quad u(1,t)=2t, \quad t\in[0,1], \quad c=1.$$

11. $$\frac{\partial u}{\partial t}=\frac{\partial^2 u}{\partial x^2}+u-x+2\sin 2x\cos x, \quad 0<x<\frac{\pi}{2}, \quad t\in(0,1],$$
$$u(x,0)=x, \quad 0\le x\le\frac{\pi}{2},$$
$$u(0,t)=0, \quad \frac{\partial u(\pi/2,t)}{\partial x}=1, \quad t\in[0,1].$$

12. $$\frac{\partial u}{\partial t}=\frac{\partial^2 u}{\partial x^2}+x\cos t, \quad 0<x<1, \quad t\in(0,1],$$
$$u(x,0)=\sin(\pi x), \quad 0\le x\le 1,$$
$$\frac{\partial u(0,t)}{\partial x}=0, \quad \frac{\partial u(1,t)}{\partial x}+u(1,t)=1, \quad t\in[0,1].$$

13. $\dfrac{\partial u}{\partial t}=\dfrac{\partial^2 u}{\partial x^2},\quad 0<x<1,\quad t\in(0,1],$

$u(x,0)=\sin(\pi x),\quad 0\le x\le 1,$

$\dfrac{\partial u(0,t)}{\partial x}=0,\quad \dfrac{\partial u(1,t)}{\partial x}+u(1,t)=1\quad t\in[0,1].$

14. $\dfrac{\partial u}{\partial t}=\dfrac{\partial^2 u}{\partial x^2}-u+x,\quad 0<x<1,\quad t\in(0,1],$

$u(x,0)=0,\quad 0\le x\le 1,$

$u(0,t)=0,\quad u(1,t)=1,\quad t\in[0,1].$

15. $\dfrac{\partial u}{\partial t}=\dfrac{\partial^2 u}{\partial x^2}+(1-t)e^{-x},\quad 0<x<1,\quad t\in(0,1],$

$u(x,0)=0,\quad 0\le x\le 1,$

$u(0,t)=t,\quad u(1,t)=te^{-1},\quad t\in[0,1].$

16. $\dfrac{\partial u}{\partial t}=\dfrac{\partial^2 u}{\partial x^2}+(1+\alpha x)^4-12\alpha^2 t(1+\alpha x)^2,\quad 0<x<1,\quad t\in(0,2],$

$u(x,0)=0,\quad 0\le x\le 1,$

$u(0,t)=t,\quad u(1,t)=t(1+\alpha)^4,\qquad t\in[0,2],\quad \alpha=1,2.$

17. $\dfrac{\partial u}{\partial t}=\dfrac{\partial^2 u}{\partial x^2}+e^{-u}+e^{-2u},\quad 0<x<1,\quad t\in(0,1],$

$u(x,0)=\ln(x+2),\quad 0\le x\le 1,$

$u(0,t)=\ln(t+2),\quad u(1,t)=\ln(t+3),\quad t\in[0,1].$

18. $\dfrac{\partial u}{\partial t}=\dfrac{\partial^2 u}{\partial x^2}-\dfrac{\partial u}{\partial x},\quad 0<x<1,\quad t\in(0,1],$

$u(x,0)=e^{x/2},\quad 0\le x\le 1,$

$u(0,t)=0,\quad u(1,t)=0,\quad t\in[0,1].$

19. $\dfrac{\partial u}{\partial t}=\dfrac{\partial^2 u}{\partial x^2}-\left(\dfrac{\partial u}{\partial x}\right)^2,\quad 0<x<1,\quad t\in(0,10],$

$u(x,0)=0,\quad 0\le x\le 1,$

$u(0,t)=1,\quad \dfrac{\partial u(1,t)}{\partial x}=0,\quad t\in[0,10],\quad N=50.$

20. $\dfrac{\partial u}{\partial t}=\dfrac{\partial^2 u}{\partial x^2}+\dfrac{1}{2}(x^2-2t),\quad 0<x<1,\quad t\in(0,1],$

$u(x,0)=0,\quad 0\le x\le 1,$

$u(0,t)=0,\quad u(1,t)=\dfrac{1}{2}t,\quad t\in[0,1].$

21. $\dfrac{\partial u}{\partial t}=\dfrac{\partial^2 u}{\partial x^2}+2x+t,\quad 0<x<1,\quad t\in(0,1],$

$u(x,0)=(1{,}1x^2+1{,}2)\sin\pi x,\quad 0\le x\le 1,$

$u(0,t)=0,\quad u(1,t)=0,\quad t\in[0,1].$

22. $\dfrac{\partial u}{\partial t}=\dfrac{1}{\alpha}\dfrac{\partial^2 u}{\partial x^2},\quad 0<x<1,\quad t\in(0,1],$

$u(x,0)=e^{-\alpha x},\quad 0\le x\le 1,$

$u(0,t)=e^{\alpha t},\quad u(1,t)=e^{\alpha(t-1)},\quad t\in[0,1],\quad \alpha=2.$

23. $\dfrac{\partial u}{\partial t}=\dfrac{\partial^2 u}{\partial x^2}+x,\quad 0<x<5,\quad t\in(0,3],$

$u(x,0)=2x,\quad 0\le x\le 5,$

$u(0,t)=1,\quad \dfrac{\partial u(5,t)}{\partial x}=2t+24,\quad t\in[0,3].$

24. $\dfrac{\partial u}{\partial t}=\dfrac{\partial}{\partial x}\left((1+t+x^2)\dfrac{\partial u}{\partial x}\right)-xu,\quad 0<x<1,\quad t\in(0,\,0{,}1],$

$u(x,0)=0,\quad 0<x<1,$

$u(0,t)+\cos t\dfrac{\partial u(0,t)}{\partial x}=t\cos t,$

$u(1,t)+(1+t^2)\dfrac{\partial u(1,t)}{\partial x}=t(2+t^2),\quad t\in[0,\,0{,}1].$

25. $\dfrac{\partial u}{\partial t}=\dfrac{\partial}{\partial x}\left((1+x^2)\dfrac{\partial u}{\partial x}\right)+2x\dfrac{\partial u}{\partial x},\quad 0<x<1,\quad t\in(0,\,0{,}1],$

$u(x,0)=0,\quad 0<x<1,$

$u(0,t)+\cos t\dfrac{\partial u(0,t)}{\partial x}=t\cos t,\quad u(1,t)=t(2+t^2),\quad t\in[0,\,0{,}1].$

Зміст

НАВЧАЛЬНЕ ВИДАННЯ

МЕТОД ПРЯМИХ ЧИСЕЛЬНОГО РОЗВ'ЯЗУВАННЯ РІВНЯНЬ ПАРАБОЛІЧНОГО ТИПУ

МЕТОДИЧНІ ВКАЗІВКИ

до лабораторної роботи з курсу

«Чисельні методи математичної фізики»

для студентів базового напрямку

6.08.02 «Прикладна математика»

Укладачі Кутнів Мирослав Володимирович
Паздрій Оксана Ігорівна

Редактор

Комп'ютерне верстання